

Resolución Taller: Neo4j Sandbox

Observación: las resoluciones pueden contener errores y deben ser consideradas únicamente de referencia. Se recomienda **encarecidamente** que intente realizar los ejercicios por **cuenta propia** antes de mirar la resolución.

Resolución

- Ej. 1. Muestre en orden alfabético el nombre, apellido y posición de los primeros 10 jugadores cuyo apellido (`family_name`) sea “Fernández” (**Obs:** ‘Fernández’ con tilde)

```
MATCH (p:Player{family_name:"Fernández"})
RETURN p.given_name, p.family_name, p.position
ORDER BY p.given_name ASC
LIMIT 10
```

- Ej. 2. Encuentre todas las finales de la base de datos. Muestre la fecha del partido, los participantes y el resultado final.

```
MATCH (m:Match)
WHERE m.stage="final"
RETURN m.date, m.name, m.score
```

- Ej. 3. Encuentre los últimos 10 partidos jugados por la selección Argentina.
(sugerencia: utilice `CONTAINS` sobre `Match.name`)

```
MATCH (m:Match)
WHERE m.name CONTAINS "Argentina"
RETURN m.date, m.name, m.score
ORDER BY m.date DESC
LIMIT 10
```

El ejercicio también se puede realizar utilizando las artistas `:IS_AWAY` e `:IS_HOME` entre los nodos `:Team` y `:Match`

- Ej. 4. Encuentre el nombre, el país y la ciudad de los estadios en los que Messi ha hecho gol.

```
MATCH (messi:Player{family_name:"Messi"})
      -[:SCORED_IN]-
      (m:Match)
      -[:PLAYED_AT]-
      (s:Stadium)
RETURN DISTINCT s.name, s.country, s.city
```

- Ej. 5. Encuentre a todos los arqueros que han metido gol en contra. Indique la fecha y el partido.

```
MATCH
      (p:Player{position:"Goalkeeper"})
      -[g:SCORED_IN{own_goal:true}]-
      (m:Match)
RETURN p.given_name, p.family_name, m.name, m.date
ORDER BY m.date ASC
```

- Ej. 6. Encuentre a los jugadores argentinos que hicieron gol el mismo partido que hizo un gol Messi.

```
MATCH
      (messi:Player{family_name:"Messi"})
      -[:SCORED_IN]-
      (:Match)
      -[:SCORED_IN]-
      (p:Player)
      -[:REPRESENTS]-
      (arg:Team{code:"ARG"})
WHERE p.family_name <> "Messi"
RETURN DISTINCT p.given_name, p.family_name
```

- Ej. 7. Encuentre todos los equipos participantes del mundial 2022 que no participaron en el mundial 2018

```
MATCH (t:Team)-[:PARTICIPATES_IN]-(:WorldCup{year:2022})
WHERE NOT ((t)-[:PARTICIPATES_IN]-(:WorldCup{year:2018}))
RETURN t.name
```

- Ej. 8. Encuentre los jugadores finalistas del mundial 2022 que nunca jugaron un partido entrando como suplente en dicho mundial.

```
MATCH (p:Player)-[:PLAYED_IN]- (m:Match{stage:"final"}),
      (m)-[:PART_OF]- (wc:WorldCup{year:2022})
```

```
WHERE NOT
(p) - [:PLAYED_IN]{substitute:true}] - (:Match) - [:PART_OF] - (wc)
RETURN p
```

- Ej. 9. Encuentre todos los pares de equipos que se hayan enfrentado más de una vez en una misma edición de la Copa del Mundo. sugerencia: utilice < para distinguir partidos/equipos para evitar duplicados

```
MATCH
(t1:Team) -- (m1:Match) -- (t2:Team) ,
(t1) -- (m2:Match) -- (t2) ,
(m1) - [:PART_OF] - (wc:WorldCup) ,
(m2) - [:PART_OF] - (wc)
WHERE t1.team_id < t2.team_id
AND m1.date < m2.date
RETURN wc.year AS year, t1.name, t2.name, m1.stage,
m1.score, m2.stage, m2.score
ORDER BY wc.year
```

Observe que no es necesario especificar la etiqueta en las aristas entre :Team y :Match, puesto que las únicas posibles son :IS_AWAY y :IS_HOME.

Adicional: quite la primera restricción del WHERE, observe ahora el resultado. Quite ambas restricciones y vuelva a observar el resultado. Responda: ¿qué está pasando?

- Ej. 10. Encuentre el top 5 máximos goleadores históricos de los mundiales

```
MATCH (p:Player) - [:SCORED_IN] - (m:Match)
RETURN p.given_name AS nombre, p.family_name AS apellido,
COUNT(*) AS partidos
ORDER BY partidos DESC
LIMIT 5
```

- Ej. 11. Encuentre el nombre, apellido, cantidad de partidos, cantidad de goles y promedio de gol de los 5 jugadores con el mejor promedio de gol de los mundiales. (sugerencia: sólo utilice partidos y goles de mundiales desde 1970)

```
MATCH (p:Player) - [:PLAYED_IN] - (:Match) -- (wc:WorldCup)
WHERE wc.year >= 1970
WITH p, COUNT(*) AS partidos
WHERE partidos >= 10 // Opcional: resultado más representativo

MATCH (p) - [:SCORED_IN] - (:Match) -- (wc:WorldCup)
```

```
WHERE wc.year >= 1970

RETURN
  p.given_name AS nombre,
  p.family_name AS apellido,
  partidos,
  COUNT(*) AS goles,
  toFloat(COUNT(*))/toFloat(partidos) AS promedio
ORDER BY promedio DESC
LIMIT 5
```

Ej. 12. Encuentre al jugador más joven en haber marcado un gol.

```
MATCH (p:Player)-[:SCORED_IN]-(m:Match)
WITH min(duration.inDays(p.birth_date, m.date).days) AS
minDays

MATCH (p:Player)-[:SCORED_IN]-(m:Match)
WHERE duration.inDays(p.birth_date, m.date).days = minDays
RETURN p.family_name AS jugador,
       m.name AS partido,

       // Opcional: edad exacta a la hora de convertir.
       duration.between(p.birth_date, m.date).years AS años,
       duration.inDays(
         // Cumpleaños más cercano
         p.birth_date +
         duration({years:duration.between(p.birth_date,
         m.date).years}),
         m.date
       ).days AS dias
```

Material:

- [Link](#) — Taller de Neo4j
- [Link](#) — Neo4j Sandbox
- [Link](#) — Manual Neo4j